
Abstract
We demonstrate the usefulness of the complementarity format for approximating optimal

saving and investment decisions in dynamic general equilibrium models.  Our objective is in part
pedagogic.  The essential equations for alternative representations of the Ramsey model are
presented in a compact and accessible format along with GAMS code as concrete illustration.
We present a new method for approximating the infinite horizon equilibria with endogenous
capital accumulation, and we demonstrate the advantages of this approach as compared with
techniques originally developed for optimal planning models.  The complementarity approach
does not require an ex ante specification of the growth rate in the terminal period, and it is
therefore suitable for models with endogenous growth or short time horizons.  We also consider
approximation issues arising in models with multiple infinitely-lived agents.  In these models,
changes in net indebtedness over a finite horizon must be estimated as part of the model in order
to obtain a precise approximation with a small number of time periods.

1.  Introduction
Approximation of infinite-horizon models has a long-standing tradition in the economics

literature.  Most of this literature deals with optimization methods, whereas we demonstrate the

usefulness of the complementarity format for approximating optimal saving and investment

decisions in dynamic general equilibrium models.  Our objective is pedagogic –  the essential

equations for a few models are presented in a compact and accessible format, along with

computer programs which concretely illustrate the models.  This approach is of interest to

applied economists due to the availability of  “off the shelf” software for processing these

models (see Rutherford, [1995][1999a]). 

There are two key issues involved in approximating an infinite horizon equilibrium for a

neoclassical growth model: (i) what is the size of the capital stock in the terminal period?, and

(ii) who owns the terminal capital stock?  In this paper we demonstrate the advantages of the 

complementarity formulation for answering these questions compared with techniques originally

developed for optimal planning models. 

We begin the paper with the classical Ramsey analysis of optimal economic growth

under certainty.  This is a natural starting point because of the generic representation of financial

markets.  The model represents a closed economy with perfect competition in all markets, a

representative consumer, and a constant rate of technological progress.  Although the model is

well studied in the economics literature (see, for example, Blanchard and Fischer [1989], and

Barro and Sala-i-Martin [1995]), analytical methods have limitations.  Numerical methods are, of

course, always required for empirical analysis of policy issues, and they can provide helpful

insights into properties of alternative formulations.  

In section 2 we formulate the Ramsey model as a primal nonlinear program in quantities,

as two different mixed complementarity problems (MCPs), and as a dual nonlinear program in

prices.  Preferences and technology are represented by utility and production functions in the

primal formulation and by expenditure and cost functions in the dual model.  The two MCP

formulations can be interpreted as first-order necessary conditions for the nonlinear

programming (NLP) models, and the complementarity problem associated with the dual

nonlinear program is essentially Mathiesen’s [1985] formulation of the Arrow-Debreu



2

equilibrium model.

In section 3 we consider methods of approximating the infinite horizon.  We present a

new method for approximating the infinite horizon equilibria with endogenous capital

accumulation, and we demonstrate the advantages of this approach as compared with techniques

based on optimization methods.  The complementarity approach does not require an ex ante

specification of the growth rate in the terminal period, and it is therefore suitable for models with

endogenous growth or short time horizons.  We illustrate in a few examples that the

complementarity formulation provides a more precise approximation of the infinite horizon

equilibrium than optimization methods. 

We also consider approximation issues arising in models with multiple infinitely-lived

agents.  In these models, changes in net indebtedness over a finite horizon must be calculated

within the model in order to obtain a precise approximation with a small number of time periods.

As illustration, in section 4 we present a Ramsey model with multiple regions, and we compare

approximation errors for formulations with and without net changes in assets over the finite

model horizon.

2.  Four Formulations of the Single Sector Ramsey Model
A familiar representation of the Ramsey model of saving and investment begins with a

single infinitely-lived representative agent.  The closed economy consists of a household with an

exogenous supply of labor over time.  One good is produced in each period using inputs of labor

and capital, and output in each period can be either consumed or invested.  There is perfect

competition in all markets and no taxes.  Individuals are assumed to have an infinite horizon, and

expectations by private agents are forward-looking and rational.  Hence, all agents have perfect

foresight because there is no uncertainty.  These assumptions imply that the optimal allocation of

resources by a central planner who maximizes the utility of the representative agent is identical

to the optimal allocation of resources in an undistorted decentralized economy. 

We present four alternative algebraic formulations of the Ramsey model, all of which

produce an identical optimal allocation of resources given common assumptions regarding

technology, preferences and initial endowments.  Each formulation offers a different perspective

into the workings of the Ramsey model.  We begin with the most familiar format (primal NLP),

and we proceed to two less familiar but convenient complementarity formats, and a dual 

optimization formulation.  We feel that by laying out a set of mathematically-equivalent

specifications, the researcher can develop basic insights into the nature of the equilibrium, which

can be crucial when the time comes to interpret policy results from more complex models. 

2.1. A Primal  NLP Formulation

The primal NLP formulation is based on an explicit representation of the utility function

for the single representative household.  The social planner maximizes the present value of

lifetime utility for the representative household:

where D is the time preference rate, Ct is aggregate consumption in year t, and u(A) is the

instantaneous utility of consumption. 

The representative agent maximizes utility subject to the constraint that output in period t

is either consumed or invested:

where Kt is capital in period t, and It is investment in period t.  Assuming strict monotonicity and

concavity of the production function, we have that fN(Kt) > 0 and fO(Kt) < 0.  It is convenient to

think of the production function exhibiting constant returns to scale in capital and a second factor

whose supply is exogenously specified.  Labeling the second factor labor, we could, for example,

represent diminishing returns to scale in capital through an underlying production function which

exhibits constant returns to scale in labor and capital, i.e.

C I f Kt t t+ = ( )

The capital stock in period t equals the capital stock at the start of the previous period

less depreciation plus investment in the previous period.  Hence, the capital stock is determined

by

where * is the annual rate of depreciation, and the initial capital stock in period t=0 is specified

exogenously. 

2.2.  A Complementarity Formulation based on Karush-Kuhn-Tucker Conditions

It is a simple matter to pose a nonlinear program as a complementarity problem: just form

the Lagrangian and differentiate.  Introducing multipliers for aggregate output and capital stock,

the above model produces the following system of first order conditions:



1 That is, total factor productivity growth at rate ( requires Harrod-neutral labor
productivity at rate  where " is the capital value share.

where pt is the output price in period t, and pt
K is the price of capital in period t.  As written, we

take explicit account of the non-negativity constraint for investment and assume that all other

variables are non-zero.  Hence, we do not specify a set of complementarity relations for the other

variables.  

2.3.  A Complementarity Formulation for Constant Returns Models

In order to exploit the complementarity format for economic equilibrium proposed by

Mathiesen [1985], we expand the class of markets represented in the model in order to treat all

production activities as constant returns to scale in model inputs.  This is possible through the

introduction of an additional primary factor, labor.  We can then define the instantaneous unit

cost function:

where rK is the rental rate of capital, and w is the real wage rate.  For example, if we assume that

total factor productivity grows at a constant rate, (, we have:

then:1

and:

This formulation further relies on the existence of closed-form demand functions which

express consumption demands as a function of market prices and income, M.  We then define:

For example, if we have logarithmic instantaneous utility, we obtain the demand function:

or, if utility is isoelastic, , then we have:

Having defined uncompensated demand functions, we can characterize the equilibrium

conditions in terms of three classes of equations: (i) zero profit conditions for all constant returns

activities, (ii) market clearance conditions for all goods and factors, and (iii) income balance

equations relating factor income to expenditure.  The zero profit conditions for production,

capital accumulation and investment are:

The market clearance conditions for capital stock, capital services, labor and output in

each period are:

An income-balance constraint relates the value of expenditure to factor earnings:

Due to homogeneity of cost and demand functions, the solution is not uniquely determined and

the model determines only relative prices.  A practical normalization in a model with one

consumer is to fix M=1 and omit the income constraint.

We assume non-negativity conditions for investment.  Because of Walras’ law and non-

negativity of prices, complementary slackness conditions, It (pt - pK
t+1) = 0, arise as a feature of

the definition of an equilibrium instead of an equilibrium condition per se.

2.4.  A Dual NLP Formulation

In order to represent the model in dual form, it is necessary that the intertemporal utility

function is linearly homogenous in consumption from period 0 to the infinite horizon.  The

restriction allows us to express indirect utility as the ratio of a function of market prices to the

present value of income.  Define the expenditure function:

It follows that the following nonlinear program has first order conditions which are equivalent to

Mathiesen’s complementarity formulation (for details, see Rutherford [1999b]):

subject to:

Observe that by Shephard’s lemma:



2 For other early papers, see Eckhaus and Parikh [1968], Chakravarty [1969], and Manne
[1970].

Associating Lagrange multipliers for the three classes of constraints with Yt, Kt, and It, it can be

seen that first order optimality conditions for the dual nonlinear program correspond to market

clearance conditions in the complementarity model.  

3.  The Terminal Capital Stock
Numerical models can only be solved for a finite number of periods.  Adjustments are

therefore required to produce a model which approximates choices over the infinite horizon.  In

this section we propose a new method for approximating the infinite horizon equilibria with

endogenous capital accumulation, and we demonstrate the advantages of this approach as

compared with techniques originally developed for optimal planning models.  The new method

is only applicable in a complementarity format, but it may also be applied through sequential

nonlinear programming.  The advantage of the new approach is that it does not require an ex ante

specification of the growth rate in the terminal period.  It is therefore suitable for models with

endogenous growth or short time horizons. 

Barr and Manne [1967] introduced an early method for approximating the infinite

horizon in optimal planning models which is still used in practice.2  The method involves an

increased weight on utility of consumption in the terminal period, and a constraint on investment

in the terminal period.  Assuming that the economy is in steady state by the terminal period T

and growing at rate (, the intertemporal utility function may be divided into two parts and

written as: 

where we define the utility weight parameter in each period as:

and

All quantities grow at the same rate in steady state.  Gross investment in the terminal

period is therefore determined by the size of the capital stock in the terminal period, the

exogenous growth rate, and the capital depreciation rate.  The constraint on investment in the

terminal period assures sufficient investment to cover growth plus depreciation:

This approximation of the infinite horizon is integrable and can be applied in either NLP or

MCP formulations of the Ramsey model.  The limitation of this approach is that there is no easy



3 Appendix A provides four different Ramsey model versions formulated in GAMS
(Brooke, Kendrick and Meeraus [1992], Rutherford [1995]). 

way to determine whether the economy will be close to a steady-state in period T.3 

Our new method for approximating the infinite time horizon relies on time-separable

utility functions.  We may then decompose the infinite horizon into two distinct optimization

problems: one problem defined over the period t=0 to t=T, and a second problem defined over

the period t=T+1 to infinity.  In a single-sector model, the two sub-problems are linked through

the capital stock in period T+1.  The finite horizon problem for the representative household is: 

subject to the intertemporal budget constraint: 

And the infinite horizon problem is:

subject to the intertemporal budget constraint:

Having decomposed the model, a good terminal approximation is one in which the

capital stock in period T+1, , is close to the optimal value in the infinite-horizon program. 



4  The use of aggregate output is not essential – investment growth could be related to
consumption or any other “stable” quantity variable in the model.  

5Application of this method in an optimal growth framework involves sequential
optimization.  Beginning with initial guess for the capital stock in the terminal period, we solve
the primal NLP version of the model, and update the value of the capital stock in the terminal
period with an iterative procedure using the constraint on the growth rate of investment in the
terminal period.  Specific programming details are provided in Appendix B. 

If we know the “true” value of the capital stock in the post-terminal period then we can calculate

the true consumption and saving paths during the transitional period.  However, after a policy

shock we do not know the “true” value of the capital stock in the post-terminal period.  It could

seem convenient to impose the long run steady-state value of the capital stock, but in that case

the model horizon should be sufficiently long to converge to the steady state.  

In a complementarity formulation we can include the post-terminal capital stock as an

endogenous variable.  As a system of equations, the extra variable requires a new equation.  For

this purpose we add an equation relating the growth rate of investment in the terminal period to

the growth rate of output:4

We emphasize that this approach is appropriate only for complementarity models where the new

constraint does not introduce a reduced cost for the variables appearing in the equation.  For this

reason, the termination method is not easily introduced in optimization models.5

The balanced investment growth constraint does not require that the model achieves the

actual steady-state growth rate in period T.  The advantage of the approach is that we do not have

to impose a specific capital stock in the post-terminal period, nor a specific growth rate in the

terminal period.  This method is therefore suitable for models with endogenous growth where the

terminal growth rate is not determined ex ante, like the model by Rutherford and Tarr [1999]. 

The termination method is illustrated in Appendix B using a Ramsey model formulated in

GAMS/MPSGE (Rutherford [1999a]). 

Figure 1 illustrates the terminal effect on investment for both terminal conditions using a



6 The model is parameterized with a value share of capital equal to 0.36, the annual time
preference rate is 5 percent, the annual steady state growth rate is 2 percent, and the annual rate
of physical capital depreciation is 7 percent.

single-sector Ramsey model.6  In this calculation the initial capital stock is reduced by 20 percent

to compare the two termination methods discussed above.  We use a model simulation over a

100 year time horizon to represent the infinite-horizon saddle point path.  We then compare

computational results for the two terminal approximation methods by solving the model with

each terminal condition for a 15 year time horizon.  The model labeled “NLP” is based on the

terminal condition by Barr and Manne, and the model labeled “MCP” is based on the state

variable targeting procedure we propose.  Deviations from the “true” saddle point path are

smaller for the model based on state variable targeting (MCP) than for the model with increased

weight on utility of consumption in the terminal period and a constraint on terminal investment

(NLP).  Several other variables exhibit deviations from related infinite horizon values, but

investment is typically the most sensitive item. 

The relationship between the “average” approximation error and the model horizon is

illustrated in Figure 2.  We define the average error as the weighted-sum of deviations from the

“true” saddle point path for investment over the full model horizon.  The weights are based on

the present discounted value of future output in the initial steady state.  Hence, the weights are

determined by the interest rate and deviations from the “true” saddle point path for investment in

the near future are weighted higher than similar deviations in the more distant future.  Figure 2

shows that the average approximation error falls with the model horizon and is significantly

smaller for the MCP model compared to the NLP model.  The MCP model with state variable

targeting can therefore obtain the same average precision with fewer periods than the NLP model

with Barr and Manne’s terminal constraint.  For example, the MCP model with state variable

targeting and a 10 year horizon produces almost the same average precision as an NLP model

with Barr and Manne’s terminal constraint and a 17 year horizon. 

The results provide a practical argument for dynamic modeling in a complementarity

format: the termination method is more precise.  Hence, a given model can include fewer periods

to approximate the infinite horizon saddle path when the state variable targeting method is used

instead of Barr and Manne’s terminal constraint. 

4.  Multiple Agents and Terminal Assets
Ownership of capital is an additional issue in dynamic models with multiple infinitely-

lived agents.  In these models, changes in net asset positions across households over a finite

horizon must be calculated within the model in order to obtain a precise approximation with a

small number of periods.  We therefore have to distinguish between the value of capital goods

and the net asset positions for private households. 

To illustrate approximation issues arising in models with multiple infinitely-lived agents,

we present and use a Ramsey model with multiple regions.  Each region is endowed with an

exogenous time path of labor and an initial capital stock, and the regions are linked through

capital and consumption goods markets.  Consumption goods are either consumed in the country

of origin or used as intermediate inputs in other regions.  Financial assets, on the other hand, are

perfect substitutes and can move freely across regions, which implies that the interest rate is

determined in the international financial market. 

The intertemporal decision problem in each region is similar to the generic Ramsey

model with a single household and can be decomposed into two distinct optimization problems:

one problem defined over the period t=0 to t=T, and a second problem defined over the period

t=T+1 to infinity.  The representative household is concerned with the optimal distribution of

consumption over time, and the two intertemporal sub-problems are thus linked via the stock of

financial assets in period T+1.  The finite horizon problem for the representative household in

region r is: 

subject to the intertemporal budget constraint: 

And the infinite horizon problem is:

subject to the intertemporal budget constraint: 

where Ar,t is the stock of financial assets in region r in period t. 

Having decomposed the intertemporal decision problem, a good terminal approximation

is one in which the net asset position in each region in period T+1, Ar,T+1,  is close to the optimal

value in the infinite-horizon program.  Our starting point is the same as before.  We exploit the

complementarity format and apply the state variable targeting procedure to determine the post-

terminal capital stock in every region, Kr,T+1.  Hence, we include the post-terminal capital stock

in every region as endogenous variables and add an equation for each capital stock that relates

the growth rate of investment in the terminal period to the growth rate of output in the given

region: 

The stock of financial assets in a given region may be different from the value of the

capital stock in that region.  We therefore have to adjust the intertemporal budget constraint over

the finite horizon to account for changes in net financial wealth.  Having determined the post-

terminal capital stock in every region, the intertemporal budget constraint for a given region is

adjusted by the difference between the region’s ownershare of global financial assets and the

value of the region’s capital stock in period T+1: 

where 2r is the ownershare of global financial assets by region r in period T+1. 

We know from the intertemporal budget constraint that the stock of financial assets in

period T+1 is equal to the difference between the present value of consumption expenditures and

labor earnings from period T+1 to infinity.  All quantities in a specific region grow at the same

rate in steady state, (r, and the ownershares of global financial assets across regions in period

T+1 can be determined by: 

The expression is simplified somewhat if all regions are on a common growth path at the end of

the model horizon.  In this case, the ownershares of global financial assets across regions can be

determined by the difference between consumption expenditures and labor earnings in period T: 



7 Mention here the basic parameter values in the model. 

Figure 3 illustrates the effects on capital flows for a model with and without adjustment

for changes in net indebtedness over the finite horizon.  We apply a two-region Ramsey model,

and the initial capital stock is reduced by 20 percent in one region.7  A model simulation over a

100 year time horizon represents the infinite-horizon saddle point path, and we compare

computational results for the two terminal approximation methods by solving the model with

each intertemporal budget constraint for a 15 year time horizon.  The model labeled “Capital

adjustment” is based on the adjusted intertemporal budget constraint over the finite time horizon,

and the model labeled “Original budget constraint” is based on the original intertemporal budget

constraint.  Figure 3 illustrates that deviations from the “true” saddle point path are smaller when

the intertemporal budget constraint is adjusted for changes in net indebtedness over the finite

time horizon compared to a model version without the adjustment.  Hence, changes in net asset

positions across households over a finite horizon must be calculated within the model in order to

obtain a precise approximation with a small number of periods. 

5.  Conclusion
We have proposed a new method for approximating the infinite horizon equilibrium with

endogenous capital formation and demonstrated the usefulness of the complementarity format

for determining optimal saving and investment decisions in dynamic general equilibrium models. 

The state variable targeting method has two advantages compared with techniques based on

optimization methods.  First, state variable targeting provides a more precise approximation of

the infinite horizon equilibria than optimization methods.  The results thus provide a practical

argument for dynamic modeling in a complementarity format, because a given model can include

fewer periods to approximate the infinite horizon saddle path when the state variable targeting

method is used.  Second, state variable targeting does not require an ex ante specification of the

growth rate in the terminal period.  This method is therefore suitable for neoclassical models

with endogenous growth where the growth rate in the terminal period is not determined ex ante. 

Figure 1.  Approximation Errors in Investment (15 Year Time Horizon).  The initial capital stock is
reduced by 20 percent, and the model is solved with each terminal condition for a 15 year time horizon.  The
model labeled “NLP” is based on the terminal condition by Barr and Manne, and the model labeled “MCP” is
based on the state variable targeting procedure we propose.   Deviations from the “true” saddle point path are
smaller for the model based on state variable targeting (MCP) than for the model with increased weight on
utility of consumption in the terminal period and a constraint on terminal investment (NLP). 

Figure 2.  Discounted Average Approximation Error.  The average error is defined as the weighted-sum of
deviations from the “true” saddle point path for investment over the full model horizon.  The weights are
determined by the interest rate and deviations from the “true” saddle point path for investment in the near future
are weighted higher than similar deviations in the more distant future.  The average approximation error falls
with the model horizon and is significantly smaller for the MCP model compared to the NLP model.  The MCP
model can therefore obtain the same average precision with fewer periods than the NLP model.  

Appendix A.  GAMS Code for Alternative Formulations of the Ramsey Model 

$TITLE Basic Data.

SET     T /1*20/
SET     TFIRST(T), TLAST(T);
TFIRST(T) = YES$(ORD(T) EQ 1);
TLAST(T)  = YES$(ORD(T) EQ CARD(T));

SCALAR
G Growth rate /0.02/,
IR Interest rate /0.05/,
K0 Capital-output ratio /3.00/,
DELTA Depreciation rate /0.07/,
Kstock Capital stock index /1.00/,
I0 Base year investment,
L0 Base year labor input,
C0 Base year consumption,
KVS Base year capital value share;

                
I0 = (DELTA + G) * K0;
L0 = 1 - K0 * (DELTA + IR);
C0 = 1 - (DELTA + G) * K0;
KVS = K0 * (DELTA + IR);

PARAMETER QREF(T) Reference quantity path,
PREF(T) Reference price path;

QREF(T) = (1+G)**(ORD(T)-1);
PREF(T) = (1/(1+IR))**(ORD(T)-1);

PARAMETER ALPHA;
ALPHA(T) = ((1+G)/(1+IR))**(ORD(T)-1);
ALPHA(TLAST) = ALPHA(TLAST) / (1-(1+G)/(1+IR));

$TITLE Ramsey Model: Barr-Manne Primal.

$SYSINCLUDE GAMS-F
$INCLUDE DATA

* Declare the production and utility functions here:

F(K) == (K/K0)**KVS * QREF(T)**(1-KVS);
U(C) == LOG(C);

POSITIVE VARIABLES  
K(T) Capital stock
C(T) Consumption
I(T) Investment;

VARIABLES
UTILITY Utility function;

EQUATIONS 
CC(T) Capacity constraint,
KK(T) Capital balance,
TC(T) Terminal condition (provides for post-terminal growth),
UTIL Discounted log of consumption: objective function;

CC(T)..         F(K(T)) =E= C(T) + I(T);

KK(T+1)..       (1-DELTA)*K(T) + I(T) =G= K(T+1);

TC(TLAST)..     I(TLAST) =G= (G+DELTA)*K(TLAST);

UTIL..          UTILITY =E= -SUM(T, ALPHA(T) * U(C(T)));

MODEL PRIMAL /CC, KK, TC, UTIL/;



A2

C.L(T)       = C0;
K.LO(T)      = 1E-4;
C.LO(T)      = 1E-4;
I.LO(T)      = 1E-4;
K.FX(TFIRST) = K0;

SOLVE PRIMAL MINIMIZING UTILITY USING NLP;

K.FX(TFIRST) = K0*0.80;
SOLVE PRIMAL MINIMIZING UTILITY USING NLP;

$TITLE Ramsey Model: Barr-Manne KKT.

$INCLUDE DATA

POSITIVE VARIABLES
P(T) Price of output,
PK(T) Price of capital,
RK(T) Price of rental capital,
PTC(T) Price of capital in terminal period
K(T) Capital stock
C(T) Consumption
I(T) Investment;

EQUATIONS
CC(T) Capacity constraint,
KK(T) Capital balance,
TC(T) Terminal condition (provides for post-terminal growth),
OPT_C(T) First order optimality condition for C,
OPT_K(T) First order optimality condition for K,
OPT_I(T) First order optimality condition for I;

CC(T).. (K(T)/K0)**KVS * QREF(T)**(1-KVS) =E= C(T) + I(T);

KK(T).. (1-DELTA)*K(T-1) + (K0*KSTOCK)$TFIRST(T) + I(T-1) =G= K(T);

TC(TLAST).. I(TLAST) =G= (G+DELTA)*K(TLAST);

OPT_C(T).. C(T)*P(T) =E= C0 * ALPHA(T);

OPT_K(T).. PK(T) + (PTC(T)*(G+DELTA))$TLAST(T) =E= 
P(T)*KVS*(K(T)/K0)**KVS * QREF(T)**(1-KVS)/K(T) + PK(T+1)*(1-DELTA);

OPT_I(T).. P(T) =E= PK(T+1) + PTC(T)$TLAST(T);

MODEL KKT / OPT_C.C, OPT_K.K, OPT_I.I, CC.P, KK.PK, TC.PTC /;

C.L(T)   = C0 * QREF(T);
I.L(T)   = I0 * QREF(T);
K.L(T)   = K0 * QREF(T);

P.L(T)   = PREF(T);
PK.L(T)  = PREF(T) * (1+IR);
PTC.L(T) = PREF(T);

PK.LO(T) = 1E-6;
PK.UP(T) = +INF;

SOLVE KKT USING MCP;

KSTOCK = 0.8;
SOLVE KKT USING MCP;

$TITLE Ramsey Model: Barr-Manne MCP. 

$SYSINCLUDE GAMS-F
$INCLUDE DATA

$if not setglobal termcnd $setglobal termcnd NLP

SCALAR NLPTERM /0/, MCPTERM/0/;
%termcnd%term = 1;

* Default data set up for NLP termination:

IF (mcpterm, ALPHA(T) = ((1+G)/(1+IR))**(ORD(T)-1));

ALIAS (T,TT);
ALPHA(T) = ALPHA(T) / SUM(TT, ALPHA(TT));

* Declare the production and utility functions here:

F(RK,PL) == (RK/(IR+DELTA))**KVS * PL**(1-KVS);
U(P) == PROD(T, (P(T)/PREF(T))**ALPHA(T));

POSITIVE VARIABLES
Y(T) Output
I(T) Investment
K(T) Capital stock
P(T) Output price
RK(T) Return to capital
U Utility
PU Unit expenditure function
PK(T) Capital price
PL(T) Wage rate
PKT Terminal capital
RA Representative agent
TK Post-terminal capital stock;

EQUATIONS
PR_Y(T) Zero profit condition for output,
PR_C(T) Zero profit condition for consumption,
PR_K(T) Zero profit condition for capital,
PR_I(T) Zero profit condition for investment,
PR_U Zero profit condition for utility,

M_P(T) Market clearing for output, 
M_PK(T) Market clearing for capital,
M_RK(T) Market clearing for rental capital,
M_PL(T) Market clearing for labor,
M_PU Market clearing for utility,
M_PKT Market clearing for terminal investment,

I_RA Income balance for representative agent,
TERMK Terminal constraint for capital stock;

PR_Y(T).. F(RK(T),PL(T)) =E= P(T);

PR_U.. U(P) =E= PU;

PR_K(T).. PK(T) + ( (PKT*(G+DELTA))$TLAST(T) )$NLPTERM =E= 
RK(T) + (1-DELTA)*(PK(T+1) + (PKT$TLAST(T))$MCPTERM);

PR_I(T).. P(T) =E= PK(T+1) + PKT$TLAST(T);

M_P(T).. Y(T) =E= ALPHA(T) * PU * U / P(T) + I(T);

M_PU.. U * PU =E= RA;

M_PK(T).. K(T) =E= (1-DELTA)*K(T-1) + I(T-1) + (K0*KSTOCK)$TFIRST(T);

M_RK(T).. K(T) * (RK(T)/(IR+DELTA)) =E= K0 * F(RK(T),PL(T)) * Y(T);

M_PL(T).. QREF(T) * PL(T) =E= F(RK(T),PL(T)) * Y(T);

M_PKT.. SUM(TLAST, I(TLAST) - (G+DELTA)*K(TLAST))$NLPTERM
+ SUM(TLAST, K(TLAST)*(1-DELTA) + I(TLAST) - TK)$MCPTERM =E= 0;

I_RA.. RA =E= SUM(T, PL(T)*L0*QREF(T)) + SUM(TFIRST, PK(TFIRST)*K0*KSTOCK)
- (TK * PKT)$MCPTERM;

TERMK$MCPTERM.. SUM(T$TLAST(T+1),  I(T+1)/I(T) - Y(T+1)/Y(T)) =E= 0;

MODEL MCP /PR_Y.Y, PR_U.U, PR_K.K, PR_I.I, 
   M_P.P, M_PU.PU, M_PK.PK, M_RK.RK, M_PL.PL, M_PKT.PKT, 
   I_RA.RA, TERMK.TK/;

Y.L(T)  = QREF(T);
I.L(T)  = I0 * QREF(T);
K.L(T)  = K0 * QREF(T);
P.L(T)  = PREF(T);
RK.L(T) = PREF(T) * (DELTA+IR);
PK.L(T) = PREF(T) * (1+IR);
PL.L(T) = PREF(T);
PKT.L   = SUM(TLAST, PREF(TLAST));
TK.L    = K0 * (1+G)**CARD(T);
PU.L    = PROD(T, (P.L(T)/PREF(T))**ALPHA(T) );
U.L     = ( SUM(T, PL.L(T)*L0*QREF(T)) + SUM(TFIRST, PK.L(TFIRST)*K0*KSTOCK)
          -  (TK.L * PKT.L)$MCPTERM) / PU.L;

RA.FX = U.L * PU.L;

Y.LO(T) = 1.E-5;
I.LO(T) = 1.E-5;

MCP.ITERLIM = 1000;
SOLVE MCP USING MCP;

KSTOCK = 0.80;
SOLVE MCP USING MCP;

$TITLE Ramsey Model: Barr-Manne Dual. 

$INCLUDE DATA

VARIABLE
OBJ Objective price;

POSITIVE VARIABLES
P(T) Price of output,
PC(T) Consumption price
PK(T) Price of capital,
RK(T) Price of rental capital,
PL(T) Wage rate,
PT Price of capital in terminal period;

EQUATIONS 
OBJDEF Objective function,
PCDEF(T) Defines PC
PRF_Y(T) Zero profit condition for output,
PRF_I(T) Zero profit condition for investment,
PRF_K(T) Zero profit condition for capital;

OBJDEF.. OBJ =E= SUM(T, ALPHA(T)*LOG(PC(T))) - SUM(T, PL(T) * QREF(T))
- SUM(TFIRST, PK(TFIRST)* K0 * KSTOCK);

PCDEF(T).. PC(T) =E= P(T) / PREF(T);

PRF_Y(T).. (RK(T)/(IR+DELTA))**KVS * (PL(T)/(1-KVS))**(1-KVS) =E= P(T);

PRF_I(T).. P(T) =E= PK(T+1) + PT$TLAST(T);

PRF_K(T).. PK(T) + (PT*(G+DELTA))$TLAST(T) =E= RK(T) + PK(T+1)*(1-DELTA);

MODEL DUAL /OBJDEF, PCDEF, PRF_Y, PRF_I, PRF_K /;

P.L(T)    = PREF(T);
PC.L(T)   = 1;
PK.L(T)   = PREF(T) * (1+IR);
RK.L(T)   = PREF(T) * (DELTA+IR);
PL.L(T)   = PREF(T) * (1-KVS);

PT.L      = SUM(TLAST, PREF(TLAST));

PC.LO(T)  = 0.001;
P.LO(T)   = 1E-6;
PK.LO(T)  = 1E-6;
RK.LO(T)  = 1E-6;
PL.LO(T)  = 1E-6;
PT.LO     = 1E-6;

SOLVE DUAL USING NLP MAXIMIZING OBJ;

KSTOCK = 0.80;
SOLVE DUAL USING NLP MAXIMIZING OBJ;

Appendix B.  GAMS Code for Alternative Termination Methods.

$TITLE Ramsey Model: State Variable Targeting MPSGE. 

$INCLUDE data

$setglobal termcnd MCP
SCALAR NLPTERM /0/, MCPTERM/0/;
%termcnd%term = 1;

$ONTEXT
$MODEL:RAMSEY

$SECTORS:
Y(T) ! Output
I(T) ! Investment
K(T) ! Capital stock

$COMMODITIES:
P(T) ! Output price
RK(T) ! Return to capital
PK(T) ! Capital price
PL(T) ! Wage rate
PKT ! Terminal capital

$CONSUMERS:
RA ! Representative agent

$AUXILIARY:
TK$MCPTERM ! Post-terminal capital stock

$PROD:Y(T) s:1
        O:P(T)          Q:1
        I:PL(T)         Q:L0
        I:RK(T)         Q:K0    P:(DELTA+IR)

$PROD:K(T)$MCPTERM
        O:PK(T+1)       Q:(1-DELTA)
        O:PKT$TLAST(T)  Q:(1-DELTA)
        O:RK(T)         Q:1
        I:PK(T)         Q:1

$PROD:K(T)$NLPTERM
        O:PK(T+1)       Q:(1-DELTA)
        O:RK(T)         Q:1
        I:PK(T)         Q:1
        I:PKT$TLAST(T)  Q:(G+DELTA)

$PROD:I(T)
        O:PK(T+1)       Q:1
        O:PKT$TLAST(T)  Q:1
        I:P(T)          Q:1

$DEMAND:RA  s:1
        D:P(T)$MCPTERM  Q:(QREF(T)*C0)  P:PREF(T)
        D:P(T)$NLPTERM  Q:C0            P:ALPHA(T)
        E:PL(T)         Q:(L0*QREF(T))
        E:PK(TFIRST)    Q:(K0*KSTOCK)
        E:PKT$MCPTERM   Q:-1 R:TK

$CONSTRAINT:TK$MCPTERM
        SUM(T$TLAST(T+1),  I(T+1)/I(T) - Y(T+1)/Y(T)) =E= 0;

$OFFTEXT
$SYSINCLUDE mpsgeset RAMSEY

Y.L(T)  = QREF(T);
I.L(T)  = I0 * QREF(T);
K.L(T)  = K0 * QREF(T);

P.L(T)  = PREF(T);
RK.L(T) = PREF(T) * (DELTA+IR);
PK.L(T) = PREF(T) * (1+IR);
PL.L(T) = PREF(T);
PKT.L   = SUM(TLAST, PREF(TLAST));
TK.L    = K0 * (1+G)**CARD(T);

RAMSEY.ITERLIM = 1000;
$INCLUDE RAMSEY.GEN
SOLVE RAMSEY USING MCP;

KSTOCK = 0.80;
$INCLUDE RAMSEY.GEN
SOLVE RAMSEY USING MCP;

$TITLE Ramsey Model: State Variable Targeting NLP. 

$SYSINCLUDE gams-f
$INCLUDE DATA

* In this code we use no adjustment of the weight on terminal consumption:

ALPHA(T) = ((1+G)/(1+IR))**(ORD(T)-1);

SCALAR KTERM Terminal capital stock;

* Declare the production and utility functions here:

F(K) == (K/K0)**KVS * QREF(T)**(1-KVS);
U(C) == LOG(C);

POSITIVE VARIABLES  
K(T) Capital stock
C(T) Consumption
I(T) Investment;

VARIABLES
UTILITY Utility function;

EQUATIONS 
CC(T) Capacity constraint,
KK(T) Capital balance,
KT Terminal capital stock,
UTIL Discounted log of consumption: objective function;

CC(T).. F(K(T)) =E= C(T) + I(T);

KK(T+1).. (1-DELTA)*K(T) + I(T) =G= K(T+1);

UTIL.. UTILITY =E= -SUM(T, ALPHA(T)* U(C(T)));

KT.. KTERM =E= SUM(TLAST, I(TLAST) + (1-DELTA) * K(TLAST));

MODEL PRIMAL /CC, KK, KT, UTIL/;

C.L(T)       = C0;
K.LO(T)      = 1E-4;
C.LO(T)      = 1E-4;
I.LO(T)      = 1E-4;
K.FX(TFIRST) = K0;

KTERM = K0 * (1 + G)**CARD(T);

SOLVE PRIMAL MINIMIZING UTILITY USING NLP;

* Experiment:

K.FX(TFIRST) = K0*0.80;

* Initial guess on terminal capital:

KTERM = K0 * (1 + G)**CARD(T);

* Evaluate the side constraint:

PARAMETER ITERLOG ITERATION LOG
DFDK Evaluation of gradient;

SET ITER /IT0*IT6/;

LOOP(ITER,

ITERLOG(ITER,"KTERM") = KTERM;

SOLVE PRIMAL MINIMIZING UTILITY USING NLP;
LOOP(TLAST(T), ITERLOG(ITER,"F") = I.L(T)/I.L(T-1) - F(K.L(T))/F(K.L(T-1)));

KTERM = KTERM * 1.05;
SOLVE PRIMAL MINIMIZING UTILITY USING NLP;

LOOP(TLAST(T), ITERLOG(ITER,"F'") = I.L(T)/I.L(T-1) - F(K.L(T))/F(K.L(T-1)));

* Take the standard Newton iterate:

ITERLOG(ITER,"DFDK") = (ITERLOG(ITER,"F'")-ITERLOG(ITER,"F"))/(0.05*KTERM/1.05);

KTERM = KTERM/1.05 - ITERLOG(ITER,"F")/ITERLOG(ITER,"DFDK");
);

Appendix C.  GAMS Code for Model with Multiple Agents.

$TITLE Ramsey Model with Multiple Regions, MPSGE.

$INCLUDE DATA

SET R /N,S/;

ALIAS (R,RR);

PARAMETER
X0(R) Base year export levels across regions,
M0(R) Base year import levels across regions,
KS(R) Capital stock index across regions;

X0(R)  = 0.1;
M0(R)  = X0(R++1);
KS(R)  = 1;

$ONTEXT
$MODEL:RAMSEY

$SECTORS:
Y(R,T) ! Output across regions
I(R,T) ! Investment across regions
K(R,T) ! Capital stock across regions
X(R,T) ! Export index across regions
C(R,T) ! Private consumption across regions
U(R) ! Intertemporal welfare index across regions

$COMMODITIES:
P(R,T) ! Output price across regions
RK(R,T) ! Return to capital across regions
PK(R,T) ! Capital price across regions
PL(R,T) ! Wage rate across regions
PM(R,T) ! Import price across regions
PC(R,T) ! Price of private consumption across regions
PU(R) ! Intertemporal welfare price index across regions
PKT(R) ! Terminal capital across regions

$CONSUMERS:
RA(R) ! Representative agent across regions

$AUXILIARY:
TK(R) ! Post-terminal capital stock across regions
TA(R) ! Terminal adjustment for changes in assets

$PROD:Y(R,T) s:1
O:P(R,T) Q:(1+M0(R))
I:PL(R,T) Q:L0
I:RK(R,T) Q:K0 P:(DELTA+IR)
I:PM(R,T) Q:M0(R)

$PROD:X(R,T)
O:PM(RR,T)$(ORD(RR) NE ORD(R)) Q:1
I:P(R,T) Q:1

$PROD:K(R,T)
O:PK(R,T+1) Q:(1-DELTA)
O:PKT(R)$TLAST(T) Q:(1-DELTA)
O:RK(R,T) Q:1
I:PK(R,T) Q:1

$PROD:I(R,T)
O:PK(R,T+1) Q:1
O:PKT(R)$TLAST(T) Q:1
I:P(R,T) Q:1

$PROD:C(R,T)
O:PC(R,T) Q:1

I:P(R,T) Q:1

$PROD:U(R) s:1
O:PU(R) Q:(SUM(T, PREF(T)*QREF(T)*C0))
I:PC(R,T) Q:(QREF(T)*C0) P:PREF(T)

$DEMAND:RA(R)
D:PU(R) Q:(SUM(T, PREF(T)*QREF(T)*C0))
E:PL(R,T) Q:(L0*QREF(T))
E:PK(R,TFIRST) Q:(K0*KS(R))
E:PKT(R) Q:-1 R:TK(R)
E:PU(R) Q:-1 R:TA(R)

$CONSTRAINT:TK(R)
        SUM(T$TLAST(T+1),  I(R,T+1)/I(R,T) - Y(R,T+1)/Y(R,T)) =E= 0;

$CONSTRAINT:TA(R)
TA(R) =E= (SUM(T$TLAST(T), P(R,T)*C(R,T) - PL(R,T)*QREF(T)*L0) / 

SUM(T$TLAST(T), SUM(RR, P(RR,T)*C(RR,T) - PL(RR,T)*L0*QREF(T) ))) *
SUM(RR, PKT(RR)*TK(RR)) - PKT(R)*TK(R);

$OFFTEXT
$SYSINCLUDE mpsgeset RAMSEY

U.L(R)    = 1;
PU.L(R)   = 1;
Y.L(R,T)  = QREF(T);
I.L(R,T)  = I0 * QREF(T);
K.L(R,T)  = K0 * QREF(T);
C.L(R,T)  = C0 * QREF(T);
X.L(R,T)  = X0(R) * QREF(T);
TK.L(R)   = K0 * (1+G)**CARD(T);
TA.L(R)   = 0;
P.L(R,T)  = PREF(T);
RK.L(R,T) = PREF(T) * (DELTA+IR);
PK.L(R,T) = PREF(T) * (1+IR);
PL.L(R,T) = PREF(T);
PC.L(R,T) = PREF(T);
PM.L(R,T) = PREF(T);
PKT.L(R)  = SUM(TLAST, PREF(TLAST));

TA.LO(R)  = -INF;

RAMSEY.ITERLIM = 0;
$INCLUDE RAMSEY.GEN
SOLVE RAMSEY USING MCP;

RAMSEY.ITERLIM = 1000;
KS("N") = 0.80;
$INCLUDE RAMSEY.GEN
SOLVE RAMSEY USING MCP;

$TITLE Ramsey Model with Multiple Regions, MCP.

$SYSINCLUDE GAMS-F
$INCLUDE DATA

ALPHA(T) = ((1+G)/(1+IR))**(ORD(T)-1);
ALIAS (T,TT);
ALPHA(T) = ALPHA(T) / SUM(TT, ALPHA(TT));

SET R /N,S/;
ALIAS (R,RR);

PARAMETER
X0(R) Base year export levels across regions,
M0(R) Base year import levels across regions,
KS(R) Capital stock index across regions;

X0(R)  = 0.1;

M0(R)  = X0(R++1);
KS(R)  = 1;

PARAMETER
CVS(R) Capital value share across regions,
LVS(R) Labor value share across regions,
IVS(R) Import value share across regions;

CVS(R) = KVS / (1+M0(R));
LVS(R) = (1-KVS) / (1+M0(R));
IVS(R) = M0(R) / (1+M0(R));

* Declare the production and utility functions here:

F(RK(R),PL(R),PM(R)) == (RK(R)/(IR+DELTA))**CVS(R) * PL(R)**LVS(R) * PM(R)**IVS(R);
UT(P(R)) == PROD(T, (P(R,T)/PREF(T))**ALPHA(T));

POSITIVE VARIABLES
Y(R,T) Output across regions,
I(R,T) Investment across regions,
K(R,T) Capital stock across regions,
X(R,T) Export index across regions,
C(R,T) Private consumption across regions,
U(R) Intertemporal welfare index across regions,

P(R,T) Output price across regions,
RK(R,T) Return to capital across regions,
PK(R,T) Capital price across regions,
PL(R,T) Wage rate across regions,
PM(R,T) Import price across regions,
PC(R,T) Price of private consumption across regions,
PU(R) Intertemporal welfare price index across regions,
PKT(R) Terminal capital across regions,

RA(R) Representative agent across regions,
TK(R) Post-terminal capital stock across regions,
TA(R) Terminal adjustment for changes in assets;

EQUATIONS
PR_Y(R,T) Zero profit condition for output,
PR_I(R,T) Zero profit condition for investment,
PR_K(R,T) Zero profit condition for capital,
PR_X(R,T) Zero profit condition for export,
PR_C(R,T) Zero profit condition for private consumption,
PR_U(R) Zero profit condition for intertemporal welfare,

M_P(R,T) Market clearing for output, 
M_RK(R,T) Market clearing for rental capital,
M_PK(R,T) Market clearing for capital,
M_PL(R,T) Market clearing for labor,
M_PM(R,T) Market clearing for export,
M_PC(R,T) Market clearing for private consumption,
M_PU(R) Market clearing for intertemporal welfare,
M_PKT(R) Market clearing for terminal capital,

I_RA(R) Income balance for representative agents,

TERMK(R) Terminal constraint for capital stock,
TERMA(R) Terminal constraint for assets;

PR_Y(R,T).. F(RK(R,T),PL(R,T),PM(R,T)) =E= P(R,T);

PR_U(R).. UT(P(R)) =E= PU(R);

PR_K(R,T).. PK(R,T) =E= RK(R,T) + (1-DELTA)*(PK(R,T+1) + PKT(R)$TLAST(T));

PR_I(R,T).. P(R,T) =E= PK(R,T+1) + PKT(R)$TLAST(T);

PR_C(R,T).. P(R,T) =E= PC(R,T);

PR_X(R,T).. P(R,T) =E= PM(R++1,T);

M_P(R,T).. Y(R,T)*(1+M0(R)) =E= C(R,T) + I(R,T) + X(R,T);

M_PU(R).. U(R) * SUM(T, PREF(T)*QREF(T)*C0) =E= RA(R) / PU(R);

M_PK(R,T).. K(R,T) =E= (1-DELTA)*K(R,T-1) + I(R,T-1) + (K0*KS(R))$TFIRST(T);

M_RK(R,T).. K(R,T) * (RK(R,T)/(IR+DELTA)) =E= K0 * F(RK(R,T),PL(R,T),PM(R,T)) * Y(R,T);

M_PL(R,T).. QREF(T) * PL(R,T) =E= F(RK(R,T),PL(R,T),PM(R,T)) * Y(R,T);

M_PC(R,T).. C(R,T) =E= ALPHA(T) * PU(R) * U(R) * SUM(TT, PREF(TT)*QREF(TT)*C0) / PC(R,T);

M_PM(R,T).. X(R++1,T) * PM(R,T) =E= X0(R) * F(RK(R,T),PL(R,T),PM(R,T)) * Y(R,T);

M_PKT(R).. SUM(TLAST, K(R,TLAST)*(1-DELTA) + I(R,TLAST) - TK(R)) =E= 0;

I_RA(R).. RA(R) =E= SUM(T, PL(R,T)*L0*QREF(T)) + SUM(TFIRST, PK(R,TFIRST)*K0*KS(R))
- TK(R) * PKT(R);

TERMK(R).. SUM(T$TLAST(T+1),  I(R,T+1)/I(R,T) - Y(R,T+1)/Y(R,T)) =E= 0;

TERMA(R).. TA(R) =E= (SUM(T$TLAST(T), P(R,T)*C(R,T) - PL(R,T)*QREF(T)*L0) / 
SUM(T$TLAST(T), SUM(RR, P(RR,T)*C(RR,T) - PL(RR,T)*L0*QREF(T) ))) *
SUM(RR, PKT(RR)*TK(RR)) - PKT(R)*TK(R);

MODEL MCP /PR_Y.Y, PR_U.U, PR_K.K, PR_I.I, PR_C.C, PR_X.X, 
   M_P.P, M_PU.PU, M_PK.PK, M_RK.RK, M_PL.PL, M_PC.PC, M_PM.PM, M_PKT.PKT, 
   I_RA.RA, TERMK.TK, TERMA.TA/;

Y.L(R,T)  = QREF(T);
I.L(R,T)  = I0 * QREF(T);
K.L(R,T)  = K0 * QREF(T);
C.L(R,T)  = C0 * QREF(T);
X.L(R,T)  = M0(R) * QREF(T);
P.L(R,T)  = PREF(T);
RK.L(R,T) = PREF(T) * (DELTA+IR);
PK.L(R,T) = PREF(T) * (1+IR);
PL.L(R,T) = PREF(T);
PC.L(R,T) = PREF(T);
PM.L(R,T) = PREF(T);
PKT.L(R)  = SUM(TLAST, PREF(TLAST));
PU.L(R)   = 1;
U.L(R)    = 1;
TK.L(R)   = K0 * (1+G)**CARD(T);
TA.L(R)   = 0;
RA.L("S") = SUM(T, PL.L("S",T)*L0*QREF(T)) + SUM(TFIRST, PK.L("S",TFIRST)*K0*KS("S"))

        -  TK.L("S") * PKT.L("S");
RA.FX("N")= SUM(T, PL.L("N",T)*L0*QREF(T)) + SUM(TFIRST, PK.L("N",TFIRST)*K0*KS("N"))

        -  TK.L("N") * PKT.L("N");

Y.LO(R,T) = 1.E-5;
I.LO(R,T) = 1.E-5;
TA.LO(R)  = -INF;

MCP.ITERLIM = 0;
SOLVE MCP USING MCP;

MCP.ITERLIM = 1000;
KS("N") = 0.80;
SOLVE MCP USING MCP;




